

OpenMP for Computational Scientists

5: Programming your GPU with OpenMP

Tom Deakin
University of Bristol

- ▶ Quick recap of NUMA exercise
- ▶ GPU introduction
- ▶ The OpenMP `target` construct
- ▶ Device data environment
- ▶ Memory movement
- ▶ Asynchronous offload
- ▶ Tools

Take your vectorised 5-point stencil code, with optimised memory access patterns, and consider NUMA issues.

- ▶ Only one change required: add `!$omp parallel do` to initialisation loops.

Dual-socket Intel Xeon E5-2680 v4 @ 2.40GHz (2x14 cores), Intel 2018, -O3 -xHost. $nx = ny = 20,000$, $ntimes = 30$. Removed `write` statement. Taken best of 5 runs.

Version	Runtime (s)	Memory bandwidth (GB/s)
Initial parallel reduction	25.667	7.48
Swap loops + vectorise	4.876	39.38
NUMA aware initialisation	2.365	81.18

2X improvement in runtime of the *main kernel*!

- ▶ 81.18 GB/s is 63% of STREAM Triad bandwidth on this Broadwell system.
- ▶ This is good! We wouldn't expect to reach Triad bandwidth for more realistic examples.
- ▶ Missing bandwidth can be explained by looking at vectorisation reports:
 - ▶ The STREAM kernels use *streaming stores*, and the 5-point stencil doesn't.



- ▶ Streaming stores write directory to main memory, avoiding cache.
- ▶ Prevents invoking Intel's read-for-ownership mechanism:
 - ▶ A write to cache required first *reading* the memory (to place it in cache),
 - ▶ and then writing to it.
 - ▶ Doubles the data movement.

- ▶ Streaming stores write directory to main memory, avoiding cache.
- ▶ Prevents invoking Intel's read-for-ownership mechanism:
 - ▶ A write to cache required first *reading* the memory (to place it in cache),
 - ▶ and then writing to it.
 - ▶ Doubles the data movement.
- ▶ STREAM Triad *without* streaming stores gets 94.6 GB/s instead of 129.0 GB/s.
- ▶ Our kernel can't use streaming stores (alignment): 85.8% achievable bandwidth.

- ▶ Streaming stores write directory to main memory, avoiding cache.
- ▶ Prevents invoking Intel's read-for-ownership mechanism:
 - ▶ A write to cache required first *reading* the memory (to place it in cache),
 - ▶ and then writing to it.
 - ▶ Doubles the data movement.
- ▶ STREAM Triad *without* streaming stores gets 94.6 GB/s instead of 129.0 GB/s.
- ▶ Our kernel can't use streaming stores (alignment): 85.8% achievable bandwidth.
- ▶ Much closer! Missing bandwidth may be gained through improving data locality in caches (via tiling).

Why use a GPU?

- ▶ Hardware trends developing highly parallel processors.
- ▶ Many simple cores vs few complex cores is one approach.
- ▶ E.g.: NVIDIA Volta GPUs offer 4.4X memory bandwidth and 1.9X the FLOPS/s of dual-socket Intel Xeon (Skylake).



- ▶ GPUs made of many cores. NVIDIA call them Streaming Multiprocessors (SMs):
 - ▶ V100 has 80 SMs.
 - ▶ P100 has 56 SMs.
- ▶ Each SM consists of 64 FP32 CUDA cores.
- ▶ CUDA cores are really organised as 2 vector units 32 wide (called warps).

Take away


GPUs are really vector-architectures made up of smaller blocks which execute together.

- ▶ GPUs are *throughput optimised*, whereas CPUs are *latency optimised*.
- ▶ Throughput optimised also called *latency tolerant*.
- ▶ GPUs achieve this by running many operations at once, and overlapping these with each other.
- ▶ Hence need many (many) operations. . .
- ▶ A V100 has 5,120 processing elements, each needing multiple units of work to overlap.

Take away


Massive amounts of parallelism to exploit.

OpenMP has a host/device model.



- ▶ Can have more than one device.
- ▶ Devices are connected to a host CPU via interconnect, such as PCIe or NVLink.
- ▶ Devices come with their own memory. On NVIDIA HPC GPUs Pascal/Volta this is HBM.

- ▶ Execution begins on the host CPU, with zero or more devices connected to the host.
- ▶ Memory spaces *not* shared!
- ▶ In OpenMP, some data copied automatically, plus controls for explicit copying.
- ▶ Directives are used to transfer execution to the device.
!\$omp target [clause [clause] ...]
!\$omp end target
- ▶ Host execution idles until target region completes (exact semantics based on tasks: see next session!).



- ▶ Programming the kernels/loops themselves is often the “easy” bit!
- ▶ Most of your programming time will be spent in getting minimal memory movement between host and device.
- ▶ Performance of the kernels themselves is often good right away assuming you’re working with code that was good on a CPU.
- ▶ Optimisations for memory layout for vectorisation often apply to GPUs.

Get code region running on the device.

```
!$omp target [clause [clause]...]
```

...

```
!$omp end target
```

- ▶ Starts executing *in serial* on the target device.
- ▶ Need other constructs to expand parallelism.
- ▶ `nowait` clause:
 - ▶ Allows host thread to continue working. Must synchronise later using tasks.
- ▶ Other clauses mainly about memory movement, which we'll come to later.

The one construct you'll need

In general, you'll run loops on the device using:

```
!$omp target teams distribute parallel do  
do i = 1, N  
    ... ! Loop body  
end do  
!$omp end target teams distribute parallel do
```

We'll walk through what the constituent parts mean.


Warning!

Not using this combined statement can have severe performance issues.

```
!$omp target teams  
... ! Code  
!$omp end target teams
```

- ▶ OpenMP *threads* on a device are grouped into a *team*.
- ▶ Can synchronise threads *within* a team.
- ▶ *Cannot* synchronise between teams (must exit **target** region for this).
- ▶ Groups of teams are called a *league*.
- ▶ **target** construct offloads (serial) execution to device.
- ▶ `teams` construct creates a league of times.
- ▶ Master thread in *each* team (redundantly) executes the code.

- ▶ The **target** teams construct creates a number of teams on the GPU, each containing one thread.
- ▶ All threads execute code block.




- ▶ Distribute iterations of a loop across teams.
- ▶ Each team gets part of the iteration space.
- ▶ Change default assignment with `dist_schedule(static)` clause. Optionally include chunk size.
- ▶ Still only the master thread in the team executes them.


```
!$omp target teams distribute  
do i = 1, N  
... ! Code  
end do  
!$omp end target teams distribute
```

- ▶ The **target** teams distribute construct distributes loop iterations to the teams.
- ▶ Teams still only contain one thread.
- ▶ Each team computes a different iteration range.


i=1,25




i=26,50



i=51,75



i=76,100




- ▶ Same semantics as on the CPU!
- ▶ Launches threads within the team and distributes iterations across those threads.
- ▶ Note, iterations that were assigned to the team by the `distribute` construct are distributed across threads in the team.
- ▶ Can use the `schedule` clause too.

```
!$omp target teams distribute parallel do  
do i = 1, N  
... ! Code  
end do  
!$omp end target teams distribute parallel do
```

Execution model: parallel do

- ▶ The **target** teams distribute parallel **do** construct launches threads in each team.
- ▶ Threads in the team share iteration space assigned by distribute construct.
- ▶ Finally have lots of parallel execution!

i=1,25




i=26,50



i=51,75



i=76,100



- ▶ The `simd` construct is also valid on the `distribute parallel do` construct.
- ▶ OpenMP says this means SIMD instructions are generated.
- ▶ Minor implementation details differ between compilers.
- ▶ Clang and IBM XL compilers ignore the `simd` clause.
- ▶ Cray ignores the `parallel do` and issues a warning about it, but does a good job of autovectorising.
- ▶ *`!$omp target teams distribute parallel do`* is a portable solution for practically obtaining the same parallelism across compilers.

Simple vector addition kernel to illustrate GPU execution.

```
!$omp target teams distribute parallel do  
do i = 1, N  
    c(i) = a(i) + b(i)  
end do  
!$omp end target teams distribute parallel do
```

Might not work out of the box as we haven't said anything about memory movement.

- ▶ Remember: memory is *not* shared between host and target.
- ▶ OpenMP uses a combination of implicit and explicit memory movement.
- ▶ This is the most complicated part of the offload specification.
- ▶ Memory movement is often a performance killer.
 - ▶ A V100 has 900 GB/s peak memory bandwidth.
 - ▶ Connected to the host via PCIe with 32 GB/s peak bandwidth.
 - ▶ Transfers between host and device are relatively very slow: minimise them.

- ▶ Data needs mapping between host and device memory spaces.
- ▶ Variable names exist in host and device space: the compiler sorts out which one you mean when you use them in your code.

```
x(i) = 1.0
```

```
!$omp target  
x(i) = 1.0  
!$omp end target
```

- ▶ OpenMP runtime and compilers must work out when `x` is in host memory or device memory.
- ▶ Like having two arrays: `h_x` and `d_x` on the host and device respectively.

- ▶ Mapping/transfers between host and device memory spaces occur when
 - ▶ enter/exit a `target` region.
 - ▶ `target` enter/exit `data` constructs.
 - ▶ update construct.
- ▶ Default behaviour:
 - ▶ Scalars are mapped `firstprivate`.
 - ▶ This means the *do not* get copied back to the host.
 - ▶ Actually saves a memory copy as passed like a subroutine argument.
 - ▶ Stack arrays are mapped `tofrom`.
 - ▶ Heap arrays are not mapped by default.

- ▶ Specify the transfer of data between host and device on a **target** region.
- ▶ Assume we have an array $A(1:N)$ and a scalar x .
- ▶ Sizes of arrays are generally known in Fortran so don't need to specify amount of data to copy.
- ▶ Can use array slicing, but slicing whole array buggy with Cray compiler.
- ▶ (you shouldn't be using assume sized arrays in your kernels anyway!)

```
!$omp target map(...)  
...  
!$omp end target
```

Direction defined from the *host* perspective.

- ▶ `map(to: A, x)`

On entering the region, copy from host to device.

- ▶ `map(from: A, x)`

On exiting the region, copy from device to host. At start of **target** region, these are uninitialised on the device.

- ▶ `map(tofrom: A, x)`

Same as applying `map(to: ...)` and `map(from: ...)`

- ▶ `map(alloc: A)`

Allocate data on the device without copying from the host. It is uninitialised. Can later be copied back to the host (with `update` etc.) as long as allocated on host too.

```
call initialise(A,B,C)

do t = 1, N
  call update(A)
  call update(B)
  call update(C)
end do
```

- ▶ Scientific codes tend to initialise data at the start, then run many kernels.
- ▶ Generally don't worry about timing of initialisation.
- ▶ Poor practice in MPI would for example gather and scatter data to reinitialise data every iteration.
- ▶ Same applies between host and device memory spaces!

- ▶ Often want to perform initial device data environment setup once, run through iterative loop, copying back at end.
- ▶ **Do not** want to copy the data every iteration! Very expensive.
- ▶ Use **target enter data** and **target exit data** constructs to control device data environment.

```
!$omp target enter data map(to: A, B, C)

do t = 1, N
  !$omp target
  ... ! E.g. Read A and B, write C
  !$omp end target
end do

!$omp target exit data map(from: C)
```

Bulk transfers happen at beginning and end, not for every **target** region in the big loop.

Vector addition with memory movement

```
! Initialise on host  
allocate(A(N), B(N), C(N))  
A = 1.0  
B = 2.0  
  
! Copy A and B to device, and allocate space for C  
!$omp target enter data map(to: A, B) map(alloc: C)  
  
! Run vector add on device  
!$omp target teams distribute parallel do  
do i = 1, N  
    c(i) = a(i) + b(i)  
end do  
!$omp end target teams distribute parallel do  
  
! Copy C back to host  
!$omp target exit data map(from: C)
```

- ▶ Often need to transfer data between host and device between different **target** regions.
- ▶ E.g. the host does something between the two regions.
- ▶ Example on next slide. . .
- ▶ Use the update construct to move the data explicitly between host and device, in either direction.
- ▶ Remember: direction is from the *host's* perspective.


```
1  !$omp target enter data map(to: A, B, C)
2  !$omp target
3  ... ! Use A, B and C on device
4  !$omp end target
5
6  ! Copy A from device to host
7  !$omp target update from(A(1:N))
8
9  ! Change A on the host
10 A = 1.0
11
12 ! Copy A from host to device
13 !$omp target update to(A(1:N))
14
15 !$omp target
16 ... ! Use A, B and C on device
17 !$omp end target
18
19 !$omp target exit data map(from: C)
```

Use the update clause with a typical halo exchange pattern.

```
1  !$omp target enter data map(...)
2
3  do t = 1, N
4      !$omp target ...
5          ... ! run kernel
6      !$omp end target ...
7
8      ! Copy latest halo data from device to host
9      !$omp target update from(halo)
10
11     ! Exchange with MPI
12     call MPI_Sendrecv(halo, ...)
13
14     ! Copy neighbour rank data to device
15     !$omp target update to(halo)
16 end do
17
18 !$omp target exit data map(...)
```

```
integer :: i, N = 1000
real(kind=8), allocatable :: A(:), B(:)
real(kind=8) :: total

!$omp target map(to: A, B) map(tofrom: total)
!$omp teams distribute parallel do reduction(+:total)
do i = 1, N
    total = total + (A(i) * B(i))
end do
!$omp end teams distribute parallel do
!$omp end target
```

- ▶ total is a scalar, so by default is mapped firstprivate.
- ▶ I.e. Each thread on the device gets its own copy.
- ▶ Importantly, it is *not* copied back to the host at the end!
- ▶ You *must* use a map clause to bring the result back.

- ▶ By default, the host thread will idle and wait for the **target** region to complete.
- ▶ The `nowait` clause causes the **target** region to be offloaded as a task.
- ▶ The host thread can continue working asynchronously with the device!
- ▶ Must synchronise using `taskwait`, or at a barrier (explicit or implicit) depending on host threading design.
- ▶ Uses the OpenMP tasking semantics.

```
1  !$omp target nowait
2  !$omp teams distribute parallel do
3  do i = 1, 10000000
4      ... ! Lots of work
5  end do
6  !$omp end teams distribute parallel do
7  !$omp end target
8  ! Host just continues because of nowait
9
10 call expensive_io_routine()
11
12 ! Wait for target task to finish
13 !$omp taskwait
```

- ▶ **Cray** provided first vendor supported implementation targeting NVIDIA GPUs in late 2015. Latest version of CCE now supports all of OpenMP 4.5
- ▶ **IBM** XL compiler suite utilises their prior work with Clang to provide OpenMP target support for NVIDIA GPUs.
- ▶ **Clang** compiler supports OpenMP 4.5 offload to NVIDIA GPUs in 7.0. Culmination of upstreaming IBM's work.
- ▶ **Intel** began support for OpenMP 4.0 targetting Intel Xeon Phi coprocessors in 2013 (version 15.0). Compiler versions 17.0+ support OpenMP 4.5 (targetting Xeon Phi).
- ▶ **GCC** 6.1 introduced support for OpenMP 4.5 targetting Intel Xeon Phi and HSA-enabled AMD GPUs. 7.0 added support for NVIDIA GPUs.
- ▶ **PGI** compilers don't currently support OpenMP on GPUs (does support OpenMP on CPUs).

www.openmp.org/resources/openmp-compilers-tools/.

- ▶ The CUDA toolkit works with code written in OpenMP 4.5 without any special configuration.
- ▶ Useful to use the profiler `nvprof`.
- ▶ Particularly useful to check it ran on a GPU! Can silently fallback to CPU execution.
- ▶ Can generate high level profiling information, a timeline, and generate data for NVIDIA's `nvvp` profiler.



nvprof ./stencil_target

==176642== Profiling application: ./stencil_target

==176642== Profiling result:

Time(%)	Time	Calls	Avg	Min	Max	Name
86.24%	424.60ms	30	14.153ms	13.776ms	14.637ms	stencil_\$ck_L49_1
9.44%	46.496ms	33	1.4090ms	895ns	24.381ms	[CUDA memcpy HtoD]
4.31%	21.242ms	32	663.82us	1.0240us	11.176ms	[CUDA memcpy DtoH]

==176642== API calls:

Time(%)	Time	Calls	Avg	Min	Max	Name
53.72%	424.69ms	31	13.700ms	1.8730us	14.641ms	cuStreamSynchronize
37.11%	293.35ms	1	293.35ms	293.35ms	293.35ms	cuCtxCreate
5.96%	47.091ms	33	1.4270ms	6.9970us	24.584ms	cuMemcpyHtoD
2.76%	21.844ms	32	682.63us	13.226us	11.304ms	cuMemcpyDtoH
0.18%	1.4557ms	1	1.4557ms	1.4557ms	1.4557ms	cuMemHostAlloc
0.17%	1.3477ms	5	269.54us	5.2270us	580.42us	cuMemAlloc
0.04%	320.63us	30	10.687us	8.6930us	43.243us	cuLaunchKernel
0.04%	317.57us	1	317.57us	317.57us	317.57us	cuModuleLoadData
0.01%	45.755us	1	45.755us	45.755us	45.755us	cuStreamCreate
0.00%	26.802us	34	788ns	283ns	4.3010us	cuEventCreate
0.00%	4.3840us	11	398ns	309ns	585ns	cuDeviceGetAttribute
0.00%	3.7440us	5	748ns	460ns	1.2540us	cuDeviceGet
0.00%	3.6880us	3	1.2290us	356ns	2.7150us	cuDeviceGetCount
0.00%	1.0500us	1	1.0500us	1.0500us	1.0500us	cuCtxSetCurrent
0.00%	1.0230us	2	511ns	193ns	830ns	cuFuncGetAttribute
0.00%	976ns	1	976ns	976ns	976ns	cuModuleGetGlobal
0.00%	957ns	1	957ns	957ns	957ns	cuMemHostGetDevicePointer
0.00%	806ns	1	806ns	806ns	806ns	cuModuleGetFunction
0.00%	604ns	1	604ns	604ns	604ns	cuCtxGetCurrent
0.00%	442ns	1	442ns	442ns	442ns	cuFuncSetCacheConfig

nvprof output



```

1  nvprof --print-gpu-trace ./stencil_target
2
3  ==176680== Profiling application: ./stencil_target
   ↳ [74/200]
4  ==176680== Profiling result:
5
   Start Duration      Grid Size      Block Size      Regs*      SSMem*      DSMem*      Size
   ↳ Throughput      Device Context      Stream Name
6  429.51ms 17.860ms      -              -              -              -              - 122.19MB
   ↳ 6.6813GB/s Tesla P100-PCIE      1              7 [CUDA memcpy HtoD]
7  447.94ms 15.150ms      -              -              -              -              - 122.19MB
   ↳ 7.8763GB/s Tesla P100-PCIE      1              7 [CUDA memcpy HtoD]
8  463.46ms 1.4080us      -              -              -              -              - 8B
   ↳ 5.4186MB/s Tesla P100-PCIE      1              7 [CUDA memcpy HtoD]
9  463.47ms 992ns      -              -              -              -              - 4B
   ↳ 3.8455MB/s Tesla P100-PCIE      1              7 [CUDA memcpy HtoD]
10 463.90ms 14.176ms      (128 1 1)      (128 1 1)      103 1.0078KB      0B -
   ↳ - Tesla P100-PCIE      1      14 stencil_$ck_L49_1 [43]
11 478.15ms 1.5360us      -              -              -              -              - 8B
   ↳ 4.9671MB/s Tesla P100-PCIE      1              7 [CUDA memcpy DtoH]
12 478.61ms 992ns      -              -              -              -              - 8B
   ↳ 7.6909MB/s Tesla P100-PCIE      1              7 [CUDA memcpy HtoD]
13 478.63ms 14.277ms      (128 1 1)      (128 1 1)      103 1.0078KB      0B -
   ↳ - Tesla P100-PCIE      1      14 stencil_$ck_L49_1 [79]
14 492.92ms 1.1200us      -              -              -              -              - 8B
   ↳ 6.8120MB/s Tesla P100-PCIE      1              7 [CUDA memcpy DtoH]
15 492.96ms 992ns      -              -              -              -              - 8B
   ↳ 7.6909MB/s Tesla P100-PCIE      1              7 [CUDA memcpy HtoD]
16 492.97ms 14.402ms      (128 1 1)      (128 1 1)      103 1.0078KB      0B -
   ↳ - Tesla P100-PCIE      1      14 stencil_$ck_L49_1 [83]
17 507.38ms 1.3120us      -              -              -              -              - 8B
   ↳ 5.8151MB/s Tesla P100-PCIE      1              7 [CUDA memcpy DtoH]
18 507.41ms 992ns      -              -              -              -              - 8B
   ↳ 7.6909MB/s Tesla P100-PCIE      1              7 [CUDA memcpy HtoD]

```

- ▶ Port your 5-point stencil code to the GPU.
- ▶ Use the `target enter/exit data` constructs to transfer data.
- ▶ Use the `target teams distribute parallel do` construct for execution.
- ▶ Print out the grid sum for every iteration:
 - ▶ Need to use reduction clause.
 - ▶ Remember to map the reduction result!
- ▶ Extra: Think about the performance compared to your CPU version.

- ▶ Can program a GPU using OpenMP with a single pragma!
!\$omp target teams distribute parallel do
- ▶ Host/device execution model.
- ▶ Data movement between host and device data regions using:
 - ▶ map clauses,
 - ▶ **target** enter/**exit data** constructs,
 - ▶ **target** update constructs.
- ▶ The need to map back result of a reduction.